

6: Financial Calculations

The Time Value of Money	1
Growth of Money I	1
Growth of Money II	2
The FV Function	2
Amortisation of a Loan	3
Annuity Calculation	5
Comparing Investments	5
Worked examples	7
Other Financial Functions	9

The Time Value of Money

On the first day of the new millennium you placed £1 000 in a bank savings account. The bank agreed to pay you interest. The amount of the initial deposit is often called the *principal* (P), and the interest paid by the bank is set by the interest *rate* (*rate*) which will be assumed is stated as a percentage figure. We will need to know the frequency of compounding – how often the bank computes the interest and adds it to your account. At the very start of your story you have £1 000 in the account. On that day, we say that the *present value* (pv) of your saving is £1 000. When you ask questions such as, *How much will be in the account in 5 years?*, you are asking about a quantity called the *future value* (fv). Suppose you left the money in the bank for five years. If the bank compounded (calculated and added the interest) at the end of each year, we would say the principal had been in the account for 5 *periods*. If, on the other hand, interest was compounded monthly, the number of periods ($nper$) would be 60. Perhaps your saving plans also have you making addition deposits of £100 at the end of each month; these we will call *payments* (pmt).

Clearly all these quantities are interlinked. If I know the present value, the rate and the number of periods, I can compute the future value. If I know the present value, the number of periods and the future value, I can (in principle) compute the rate. Some of these calculations may be done in Excel in what could be called the step-by-step way; all of them may be performed using financial functions. We start with two examples of the step-by-step way.

Growth of Money I

If I invest £100 at 6.25% pa compounded annually, what will I have at the end of five years. This problem is set up in the worksheet shown in Figure 1.

2 Financial Functions




Figure 1

Enter the year values in column A as shown. Enter the rate of 6.25% in B2 and the initial value of 100 in B4. The formula in C2 to compute the interest is `=B4*B2` (the \$ symbol allows us to copy the formula later.) The formula in D4 is `=B4+C4`. Now we have the value of the investment at the end of year. Since the value on 1 January of the next year will be the same as that on December 31 of the previous year, enter in B5 the formula `=D4`. Copy C4:D4 down to row 5. Now we have the value at the end of the second year. The process repeats itself, so copy B5:D5 down to row 9. The value in D9 is the answer.

A plot of the values in column D against years in column A looks like a linear plot. This is misleading because the time period is so short for the low rate of interest. You can see this if you raise the rate to some large value such as 25%. The plot is then a curve - it is in fact an exponential curve.

Growth of Money II

In Figure 2 we have changed the scenario slightly. Rather than a single deposit, this time we will make a deposit of £100 at the end of each year. All we need do is modify the formula in D4, replacing `=B4 + C4` by `=B4 + C4 + 100`, and copy this down to D9. At the end of the sixth year we will have £845.81.




Figure 2

The FV Function

We could have found the answer to the first problem above in one step using the equation $A = P(1 + R)^n = 100(1 + 0.065)^6$ – see *Quantitative Approaches in Business Studies* page 396. The answer for the second problem would be more difficult. However, the Microsoft Excel functions are relatively easy to use.

The future value FV function is the one needed for the two problems above. The syntax of this is **FV(rate, nper, pmt, pv, type)**. We have already noted that the quantities *present value*, *future value*, *payment*, *number of periods* and *rate* are interrelated. For this reason, the arguments for the function are the remaining quantities. In addition, we have the *type* arguments. This has the value 0 when payments are made at the end of the period and 1 when they are made at the start. A value of 0 is assumed if the argument is omitted.

When using this, and other financial functions, care must be taken with the signs used for the monetary quantities. It is normal to use positive amounts when the cash flows in and negative amounts when the cash flows out. In the scenarios of the two problems above, the initial deposit and the subsequent annual payments flow out from the saver into the bank. From the saver's point of view, these are negative quantities. Note that when Excel computes a negative value for the financial functions the cell is formatted to show the amount in red.

The two problems from above are solved in the worksheet shown in Figure 3. Only G4 and G5 contain formulas and these are: =FV(B4,C4,D4,-E4) and =FV(B5,C5,D5,E5), respectively. In each case the *present value* (and in the second formula, the *payment*) is made negative using different approaches. In line 4, the *pvt* cell (E4) holds positive value but the formula in G4 uses -E4. In line 5, the two quantities that are to be treated as negative amounts in G5's formula are entered as negative values in their respective cells (D5 and E5).

	A	B	C	D	E	F	G
1	Growth of Money Problems						
2							
3		rate	nper	pmt	pv	type	fv
4	Problem 1	6.25%	6	0	100	0	\$143.87
5	Problem 2	6.25%	6	-100	-100	0	\$845.81

Figure 3

Amortisation of a Loan

A company borrows £10,000 to purchase a machine. The bank expects this to be paid off in monthly installments over 3-years with interest compounded quarterly at the rate of 2.5% per month. What will be the payments? How much of the principal will be paid off by the end of the first year?

The syntax for the PMT function is **=PMT(rate, nper, pv, fv, type)**. The first three arguments are the interest rate, the number of periods, and present value (amount of loan); these three are required. The optional fourth argument (*fv*) is the future value of the outstanding loan at the end of the repayment period; normally we wish this to be zero. The last option (*type*) is as before. So the formula in B6 is =PMT(B4,B5,B3) where the negation sign is used to force a positive value.

In the table from which the chart is made the formulas in row 9 are:

B9: =B3
 C9: =B9*\$B\$4 (The \$ symbols enable us to copy this formula later)
 D9: =\$B\$6 - C9 (The payment less the interest due)

4 Financial Functions

In row 10 we have:

B10: =B9-D9

C10: =B10*\$B\$4 (Copied from C9)

D10: =\$B\$6 - C10 (Copied from D9)

Row 9 is copied down to row 44.




Figure 4

Some points to note:

- In the initial months, most of the payments go to pay interest; only after month 16 does more than 50% of the payment go towards paying off the principal.
- After month 12 (i.e. at the start of the 13th month) the principal is 7 592.
- At the start of the last month, 416.16 is owed. The interest on this at month-end is 10.35 and the remainder of the payment is exactly the amount needed to reduce the principal to zero.

Annuity Calculation

An annuity makes periodic payments over a fixed length of time. Normally, one purchases an annuity for a lump sum. Suppose I wish to purchase an annuity that will give me \$3,000 each month (*payment*) for 15 years. In addition I wish the annuity to make a final payment of \$5,000 at the end of this period (this is the *future value*). How much must I pay for this annuity if the current interest rate is 1.0% per month? This is a *present value* problem since it asks about the value today of money to be received in the future. The syntax of the PV function is **PV(rate, nper, pmt, fv, type)**. Note how this parallels the FV function. The

problem is solved in the worksheet shown in Figure 5 in which B7 has the formula =PV(B4, B3*12, B5, B6, 0). The result is in red since I must pay the bank (outflow of cash from me) this amount if I am to receive (inflow of cash) the payments. If a banker were making this calculation she may make the payments negative and produce a cash inflow from the buyer of the annuity. Note the second argument (*nper*). Because it is easier to think in terms of years, B3 contains a year value but when the compounding of interest is done monthly I must work in units of months. Hence the second argument is B3*12.

	A	B	C	D	E	F
1	Annuity					
2						
3	Years	15				
4	Rate	1.00%	pm			
5	Payment	3,000				
6	fv	5,000				
7	pv	-\$250,798.91				

Figure 5

Comparing Investments

In *Quantitative Approaches in Business Studies* (page 406) two investments are compared. The initial costs and the annual revenues produced by the two investments for a three year period are shown in Figure 6.

	A	B	C	D	E	F	G	H	I	J
1	Comparing Investments									
2										
3	Cash Flow				Discount rate 15%					
4	Year	A	B		Present value calculations			Discount factors		
5	0	-4,000	-3,900		Year	A	B			
6	1	2,000	1,500		0	-4,000	-3,900			
7	2	2,500	2,500		1	1,739	1,304		0.869565	
8	3	1,500	2,000		2	1,890	1,890		0.756144	
9					3	986	1,315		0.657516	
10	Present value	616	610		Present value	616	610			
11										

Figure 6

We may calculate the present value of the cash flow using the same method as the text book. This is done in the area E2:G11. The discounted worth of each year's revenue is computed from $amount \times (1 + rate)^{-period}$. Thus in F6 we use =B6*(1+\$G\$2)^-\$E6. The \$ symbols, of course, are there to enable us to copy the formula down and across. With the discounted cash flows for project A in F5:F8 we calculate the present value for the project in F10 with =SUM(F5:F8)

Note that our values are slightly different from those in the textbook. This results from using discount factors rounded to three decimal places. In column I we show more accurate discount factors. The formula in I6 is =(1+\$G\$2)^-E6.

There is a shorter way to obtain these answers in Excel. We cannot use the PV function because the annual cash flows are not the same. We can, however, use the NPV (net present value) function. The formula in B10 is =NPV(\$G\$2,B6:B8)+B5. The first term computes the NPV of the revenues using NPV which has the syntax =NPV(rate, value1, value2,...) where Value 1, value2, etc. are the cash flows for periods 1, 2, etc. We have

6 Financial Functions

chosen to use a range for *Value1*. We must allow for the zero year cash flow – in this case the cost of the machine. This is the second term in the formula in B10. Note that since we chose to enter this as a negative value in B5, we add B5's value.

A second way of making a decision between two competing projects is to compute the *internal rate of return* for each and opt for the one with the highest value. We show these calculations in Figure 7. The formula in B14 is =IRR(B5:B8,10%). There is no simple algebraic formula to find the internal rate of return. So Excel uses an iterative process. For this to work, it needs an initial guess. This is the 10% argument in the formula. The IRR for project A exceeds that for project B – we expected as much since A has the higher present value.

	A	B	C	D	E	F	G
13	Internal rate of return				Modified internal rate of return		
14	IRR	24.4%	23.7%		MIRR	20.6%	20.7%
15							
16	FV of reinvested cash						
17	1	3,097	2,297				
18	2	3,111	3,094				
19	3	1,500	2,000				
20	Total	7,708	7,391				
21	Rate	24.4%	23.7%				

Figure 7

Many authors have commented on the inappropriateness of the IRR computation. It assumes that the revenues generated by the project can themselves be invested at the same rate as generated by the project. This is not necessarily true. We may have a marvellous machine making high-value items at low cost. Thus we are getting a good return on the investment but we may not be able to find something with the same high return in which to invest the revenues.

The calculations in A17:C21 make this point. In A17 we compute the future value of project A's the first-year revenue assuming it is invested at 24.4% for two years. The formula in A7 is =B6*(1+B\$14)^(3-\$A17). Note that the term (3-\$A17) computes the number of years the money will be invested. This formula is copied down to row 19 to compute the next two amounts. Using =SUM(B17:B19) we compute the total future value in B20. In B21 we show that this total accumulated from an invested of 4,000 does indeed required a rate of return of 24.4%. To calculate this quantity we use the geometric growth equation: $rate = \sqrt[t]{income / investment} - 1$ where t is the number of years. How do we compute cube roots (or higher roots) in Excel? Recall that $\sqrt{x} = x^{\frac{1}{2}}$, so $\sqrt[3]{x} = x^{\frac{1}{3}}$ and $\sqrt[n]{x} = x^{\frac{1}{n}}$. The formula in B21 is, therefore =(B20/-B5)^(1/3)-1.

Because of the limitations of the IRR concept, Excel also provides the *modified internal rate of return* function MIRR which allows the user to specify both the rate for the initial investment and the rate for the annual revenues. Using 15% for each of these, the MIRR values for the two projects are computed in F14 and G14. The interested reader should look in Excel's Help facility for more details.

Worked examples

On page 410 of *Quantitative Approaches in Business Studies* you will find three worked examples of financial calculations. We will see how to solve these using Excel.

1) *What is the minimum amount you would expect to pay in return for a guaranteed income of £5 000 a year for ten years, the first payment to be made at the end of year one? Assume an interest rate of 5%.*

We begin by reproducing the method in the text book – compute the present value for each annual payment using the discount formula and sum them to get the present value of this annuity. The formula in B11 is $=B\$6*(1+\$B\$5)^{(-B10)}$ and this is copied to K11. The SUM function is used in L11 to give the same value as in the textbook.

We compute the same quantity in B13 rather more quickly with $=PV(B5,10,B6)$. Note that we use only the three required argument: rate, payments and number of periods. There is no future value in this problem so we omit the fourth argument. The fifth argument specifies if payments are paid/received at the end or beginning of each period. When this is omitted, Excel assumes a value of) – payments are at the end of each period. The PV function automatically formats the cell. In this case we get a red negative value indicating that if we are to get positive payments in the future, we should expect an outflow of cash today.

	A	B	C	D	E	F	G	H	I	J	K	L
1	Worked Examples											
2												
3	Annuity question											
4												
5	Rate	5%										
6	Payment	5000.00										
7												
8												
9												
10	Year	1	2	3	4	5	6	7	8	9	10	Total
11	Discounted payment	4761.90	4535.15	4319.19	4113.51	3917.63	3731.08	3553.41	3384.20	3223.04	3069.57	£38,608.67
12												
13	Using PV	-£38,608.67										

Figure 8

2) *Projects A and B both cost £30 000. Project A yields a cash flow of £6 000 annually for 3 years, while B simply yields a lump sum of £20 000 at the end of a three-year period. Assuming a 5% interest rate, which is the more profitable project?*

Figure 9 show a worksheet set up to solve this problem. The formula in C21 is $=C20*(1+\$C\$17)^{(-C\$19)}$. This computes the discounted value of an amount after so may years. Copy this C21:F21 and to C24:F24. In column G add the discounted amounts. Project B has the least negative PV. Hopefully the company can find a project with a positive present value!

The same values may be computed using the NPV function. In I20 the formula is $=NPV(C17,D20:F20)+C20$. Because the revenues are the same in all three years for project A, one can use a similar formula with PV in place of NPV. This is left for the reader to do.

8 Financial Functions

	A	B	C	D	E	F	G	H	I	J
17	Project Decision	Rate	5%							
18										
19		Year	0	1	2	3			NPV	PV formula
20	Project A	Cash	-30000	6000	6000	6000	Total		-\$13,661	-\$13,661
21		Discounted	-30000	5714	5442	5183	-13661			
22										
23	Project B	Cash	-30000	0	0	20000	Total		NPV	
24		Discounted	-30000	0	0	17277	-12723		-\$12,723	

Figure 9

3) If £200 is invested at 7% per annum compound interest, how years will it take for the account to be worth £500?

As indicated in *Quantitative Approaches in Business Studies*, one might solve this problem using trial and error. Set up your worksheet as shown in Figure 10. The formula in B10 could be $=B30*(1+B31)^B32$ which corresponds to $200 \times (1.07)^n$ or we could use the Future Value function as in $=FV(B31,B32,0,-B30)$. In the later case note that we have -B30 since B30 was entered as a positive value but it was an outflow of cash (the depositor gave it to the bank.) Now we experiment with various values in B32 until the future value is close to the required 500. You may even wish to let Solver do this for you as explained in Unit 3.

	A	B	C	D	E	F	G	H
29	Savings problem							
30	Principal	-200			Logarithm method		Using NPER	
31	Rate	7%			Target	500		
32	Years	14			n	13.5428		
33	Future value	\$515.71						

Figure 10

Alternatively, if the future value is computed with:

$$\text{principal} \times (1 + \text{rate})^n = \text{future value}$$

then

$$(1 + \text{rate})^n = \text{future value} / \text{principal}$$

and, by taking logarithms

$$n \times \text{LOG}(1 + \text{rate}) = \text{LOG}(\text{future value} / \text{principal})$$

or

$$n = \frac{\text{LOG}(\text{future value} / \text{principal})}{(1 + \text{rate})}$$

In E13 this is entered as $=\text{LOG}(E31/B30)/\text{LOG}(1+B31)$ and yields a value of about 13½ years.

Of course, Excel provides a direct way of getting this answer. The formula in G31 is $=\text{NPER}(B31,0,-B30,E31)$. Again, we negate the B30 value to indicate an outflow of money. When you are using only Excel functions, it is better to enter all outflows as negative values and then you need not adjust the arguments in Excel functions.

Other Financial Functions

Excel has over 50 financial functions. To see a full list with a short description of each in the Help facility, in the *Answer Wizard* search box enter the name of any financial function (such as PV). Open the Help area for the function and click on *See Also*. In the *Topic Found* box, click on *financial functions*. You might expect a more direct route!